
Решайте задачи, побольше и разные – абстрактные и содержательные, «на 5 минут» и «на день работы». Все это обязательно пригодится: поверьте, что много решенных задач – не бывает.

24Задачи, подводящие к решению некоторых рассмотренных задач




ВВЕДЕНИЕ
Одним из важных условий, выполнение которых позволит вам стать хорошим программистом, является большой опыт решения задач по программированию. 
Пойа писал, что решение задач – это практическое искусство, подобно плаванию, или катанию на лыжах, или игре на пианино: вы можете научиться этому, только практикуясь. Он сравнивал, что если вы захотите научиться плавать, то вынуждены будете зайти в воду, а если вы захотите стать человеком, хорошо решающим задачи, вы вынуждены их решать. 

Историческая справка Джордж Пойа преподавал математику в Стэндфордском университете (Калифорния). Он известен своими статьями о решении задач. Однажды он написал: «Пытаясь решать задачи, вы должны наблюдать и повторять то, что делают другие, когда решают задачи, и в конце концов вы учитесь решать задачи, решая их».

Общая схема решения задачи

1. Чтение условия


На этом этапе необходимо внимательно прочесть условие, не пропуская ни одной фразы. Типичные проблемы:

· в обстановке соревнований сложно быть внимательным. Отведите достаточно времени на спокойное чтение условия. Отдохните полминуты, чтобы сконцентрироваться, но не спешите читать условие «наискосок». Неверное понимание условия может привести к тому, что вы будете решать совершенно другую задачу, а не ту, что сформулирована;

· обычно в условии есть так называемое «литературное введение», придающее задаче сюжет. Чтение такого описания обычно утомляет, отвлекает, а также расслабляет, так как авторам задач обычно не чуждо чувство юмора. Однако будьте осторожны: во введении может, прямо или косвенно, содержаться важная информация, касающаяся условия. Обычно это делают не очень внимательно, выискивая начало содержательного текста, который скорее всего пойдет потом без перерыва.

Будьте внимательны, ошибки при чтении условия дорого обходятся.

2. Построение математической модели.

 На этом этапе необходимо понять, в чем заключается задача – построить ее математическую модель «в голове». Это означает достаточно формально и математически строго понять условие.

Понять условие – это, как правило, научиться вручную, с помощью ручки и листа бумаги, находить ответ для простых наборов входных данных (тестов). Кстати, полезно не только научиться, но и проделать это для нескольких таких тестов.

3. Построение общей схемы решения.

Теперь следует перейти от понимания того, что необходимо сделать, к пониманию того, как это сделать. 

· Решение большого количества задач можно получить, разбив задачу на подзадачи («кирпичики»)

· Не все задачи решаются построением из «кирпичиков». Решение задачи может базироваться на математических идеях, которые необходимо просто придумать. 

· При решении сложной задачи, к которой никак не подступиться, имеет смысл некоторое время «продолбить» ее стандартными методами. Это может не привести к решению, а может привести и к неправильному решению, внешне похожему на правильное. Однако такой подход оправдан хотя бы потому, что большинство задач только на первый взгляд – сложные. 

4. Реализация
На этом этапе собственно пишется программа. Иногда предпочтительнее программирование «сверху вниз», иногда – «снизу вверх», или их комбинация.

5. Тестирование и отладка
Первое правило тестирования – проверяйте задачу на тесте (выборе входных данных) из примера. Далее, не ленитесь придумывать свои тесты. Вводите много «маленьких» тестов.

Второе правило – внимательно проверяйте, что программа выдала на тесте. Очень часто, когда программа правильно работала на девяти тестах, придуманных тобой, но выдает неправильный ответ на десятом, ты этого не замечаешь, поскольку запускаешь программу на десятом тесте только для «очистки совести». Кроме маленьких тестов, необходимо всегда проверять решение на так называемом «максимальном тесте».
Некоторые горячие клавиши,

которые удобно использовать при работе с редактором Pascal и отладке программы.
	Клавиша
	Что делает

	Ctrl+F7
	Поставить переменную в окно просмотра.

	F6
	Перейти в другое окно.

	F5
	Развернуть/свернуть окно.

	Alt+F3
	Закрыть окно.

	F2.
	Сохранить программу

	F9
	Проверить на ошибки

	Ctrl+F9
	Запуск программы на выполнение

	Alt+F5
	Проверка результатов выполнения программы

	F8
	Выполнить одну строку.

	F4
	Выполнить до курсора.

	Ctrl+F2
	Сбросить выполнение.

	Alt+Tab
	Переход между окнами

	Alt+x
	Выход из Pascal

	Alt+Enter
	Развернуть окно Pascal


I. Олимпиада для 1-8 классов
1. Заезд в пионерский лагерь  ( 10 баллов )
В пионерском лагере сформировали 4 отряда. В первом отряде N мальчиков (0<N<20, N- задано), а девочек в два раза больше чем мальчиков. Во втором отряде детей на 5 больше, чем в первом. В третьем отряде столько же, сколько в первом и во втором вместе взятых. В четвертом отряде детей на 10 больше, чем девочек в первом отряде. Сколько всего детей в лагере.

Пример:

Ввод:


N=6

Вывод:

104

Текст программы
program zaezd_1;{решение влоб}
 var n,otr1,otr2,otr3,otr4:integer;

begin

write('n=');

readln(n);

otr1:=n+2*n;

otr2:=otr1+5;

otr3:=otr1+otr2;

otr4:=2*n+10;

n:=otr1+otr2+otr3+otr4;

writeln(n);

end.
Преобразуем вручную выражение 

n:=otr1+otr2+otr3+otr4,

 подставляя в него исходные значения переменных otr1, otr2, otr3, otr4: 

n:=n+2*n+ n+2*n+5+ n+2*n+ n+2*n+5+ 2*n+10 

и получаем выражение

n:=14*n+20, 

результат вычисления которого и является ответом на вопрос задачи.
program zaezd_1;{решение после некоторого размышления}

 var n:integer;

begin

write('n=');

readln(n);

n:=14*n +20;

writeln(n);

end.
Тесты к задаче
	№ теста
	Вход
	Выход
	Баллы

	1
	4
	76
	2

	2
	7
	118
	2

	3
	10
	160
	2

	4
	13
	202
	2

	5
	19
	286
	2


2. Спортивный праздник ( 15 баллов )

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 

	 
	 
	 
	 
	 
	 
	 
	 
	 


При подготовке к соревнованиям необходимо покрыть всю поверхность пола спортивного зала специальными прорезиненными ковриками. Коврики квадратные и укладываются плотно друг к другу. Если коврик целиком не вмещается, то отрезается необходимая его часть. Известен размер спортивного зала: А*В метров (5<=A, D<=100 метров - целые) и размер коврика С (1<=C<=10 - целое). Определить сколько всего ковриков или отдельных кусков будет уложено.

Пример:


Ввод:


А=9, В=13, С=6

Вывод:

6
Текст программы
program prazdnik;

    var a,b,c,k,m:integer;

begin

{ввод исходных данных}

write('a=');read(a);

write('b=');read(b);

write('c=');readln(c);

{подсчет целых ковриков на стороне а}

k:=a div c;

{добавляем кусочек, если он есть}

if a mod c>0 then k:=k+1;

{подсчет целых ковриков на стороне b}

m:=b div c;

{добавляем кусочек, если он есть}

if b mod c >0 then m:=m+1;

{подсчет общего числа ковриков}

k:=k*m;

{вывод полученного результата на экран}

writeln(k);

end.
Тесты к задаче
	№ теста
	Вход
	Выход
	Баллы

	1
	5  5  1
	25
	2

	2
	6  5  2
	9
	2

	3
	10  18  3
	24
	2

	4
	10  21  4
	18
	4

	5
	100  98  3
	1122
	5


3. Пирамида ( 20 баллов )

В строительстве пирамиды на самый верхний уровень укладывается 1 камень. Уровнем ниже лежит уже 4 камня (2*2). Еще ниже – 9 камней (3*3), и т.д. На каменоломне вытесали N камней (10<N<=1 000 000 000). Какой максимальной высоты можно построить пирамиду из этих камней. Примечание: все уровни в пирамиде должны быть завершены.



Пример:

Ввод:


N=37

Вывод:

4

(Действительно: 1+4+9+16=30 камней. 7 камней останется)

Идея решения

Суммируем квадраты целых чисел до тех пор, пока не получим сумму, превышающую заданное число N. Предпоследнее целое число, которое возводится в квадрат для вычисления этой суммы и есть ответ на вопрос задачи.

Текст программы
program piramida;

   var i,n,s:longint;

begin
{ввод количества камней}

   write('n=');

   readln(n);

{поскольку n>10, то

высота пирамиды i минимум 2 уровня,

а камней для этого s нужно 5}

   i:=2;

   s:=5;

{начиная с 3-его уровня возводим i в

квадрат и добавляем к сумме.

Выходим из цикла, когда общая сумма

превысит заданное число камней }

   repeat
     inc(i);

     s:=s+sqr(i);

   until s>n;

{выводим результат}

   writeln(i-1);

end.
Тесты к задаче
	№ теста
	Вход
	Выход
	Баллы

	1
	33
	4
	2

	2
	200
	7
	3

	3
	19100
	38
	4

	4
	514123
	115
	5

	5
	333 333 333
	999
	6


4. Посчитайте вагоны ( 25 баллов )
Мимо железнодорожной станции прошло три поезда. Количество пассажиров в поездах известно: N1, N2, N3, соответственно (100<N1, N2, N3<1000). Определите, сколько пассажирских вагонов в каждом поезде, если известно, что в каждом вагоне по одинаковому числу пассажиров число их наибольшее из всех возможных.
Пример:


Ввод:


N1=120


N2=200


N3=180

Вывод:

6,10, 9

Идея решения

Проанализировав условие задачи, легко прийти к выводу, что формально задача сводится к поиску наибольшего общего делителя трех чисел, чтобы определить количество пассажиров в одном вагоне. Зная количество пассажиров в одном вагоне, легко определить количество вагонов в каждом поезде  делением общего числа пассажиров на  количество пассажиров в одном вагоне.

 Из математики известно что НОД(a,b,c)=НОД(НОД(a,b),c). Таким образом нужно дважды применить алгоритм Евклида для поиска наибольшего общего делителя двух чисел или оформить этот алгоритм в виде функции и дважды обратиться к ней в программе.
Информация для непосвященных. Алгоритм Евклида заключается в том, что мы сравниваем два числа и на место большего из них ставим остаток от деления большего на меньшее. Процесс продолжаем до тех пор, пока одно из чисел не превратиться в ноль, а другое следовательно и будет наибольшим общим делителем. Поскольку одно из чисел ноль, то НОД=сумме этих чисел .
Текст программы
program vagon;

   var n1,n2,n3,a,b,c:integer;

begin
{вводим количество пассажиров в поездах}

  write('n1='); readln(n1);

  write('n2='); readln(n2);

  write('n3='); readln(n3);

{запоминаем количество пассажиров, чтобы их не "потерять"}

  a:=n1; b:=n2; c:=n3;

{применяем алгоритм Евклида для вычислени НОД(а,b)}

  while (a>0) and (b>0) do

    if a>b then a:=a mod b

           else b:=b mod a;

{НОД(а,b) - в переменную а}

  a:=a+b;

{применяем алгоритм Евклида для вычислени НОД(а,с)}

  while (a>0) and (c>0) do

   if a>c then a:=a mod c

          else c:=c mod a;

{НОД(а,с) - в переменную а}

  a:=a+c;

{вычисляем количество вагонов в каждом поезде}

  n1:=n1 div a;

  n2:=n2 div a;

  n3:=n3 div a;

{выводим полученный результат на экран}

  writeln(n1,' ',n2,' ',n3);

end.
Тесты к задаче
	№ теста
	Вход
	Выход
	Баллы

	1
	45   45  45
	1  1  1
	1

	2
	200  160  220
	10  8  11
	6

	3
	407  333  481
	11  9  13
	6

	4
	328  246  451
	8  6  11
	6

	5
	315  180   675
	7  4  15
	6


II. Олимпиада для 1-11 классов
1. Баобаб 

Ивану Ивановичу подарили чудо-дерево, на котором рос один чудо-плод, а каждое утро должно появляться еще по одному плоду. Итак: в первый день был один чудо-плод, на 2-ой день будет 2 плода, на 3-й день – 3 плода и т.д. Поразмыслив, Иван Иванович решил продавать эти плоды на рынке. Он выяснил, что курс валют меняется каждый день,  поэтому каждый день может изменяться стоимость одного плода. 


Имеется прогноз стоимостей одного плода на N дней вперед. Требуется написать программу, которая поможет Ивану Ивановичу определить, в какие дни выгоднее всего продавать чудо-плоды для получения максимальной прибыли по истечении N дней.

Формат ввода:
В текстовом файле ВАОВАВ.IN записаны:

в первой строке натуральное число N (1
[image: image1.wmf]£

N

100), в каждой из последующих N строк – натуральное число, соответствующее стоимости С [i ] одного плода в i-й день, i=1,2…, N(1
[image: image3.wmf]£

C [i ] 

1000).

Формат вывода:
Программа должна записать в текстовый файл ВАОВАВ. OUT максимальную денежную сумму, которую Иван Иванович может заработать за N дней.

	Пример входного файла
	Пример выходного файла
	Комменарий

	3

10

15

7
	37
	На 2-й день – 2 плода и на 3-й день – 1 плод (15*2+7*1=37)


Идея решения

Для того, чтобы решить эту задачу, нужно найти максимальную стоимость плода в исходном массиве и продать по этой стоимость возможное количество баобабов. Затем исключить из рассмотрения все стоимости до дня продажи и снова найти максимальную стоимость плода в оставшейся части массива и снова продать возможное количество баобабов. Полученную от продажи сумму прибавить к предыдущей. Так продолжать до тех пор, пока максимальная стоимость плода не окажется последним элементом в массиве стоимостей.
Текст программы 
program baobab;

var j,i,pred,maxi,max,n:integer;

    s:longint;

    f1,f2:text;

    a:array[1..100] of integer;

begin

  assign(f1,'_10.in');

  reset(f1);

  assign(f2,'baobab.out');

  rewrite(f2);

{считываем в переменную n количество дней, на которые известен прогноз}  

  readln(f1,n);

{считываем в массив цену продаж по дням}

  for i:=1 to n do readln(f1,a[i]);

  i:=1;

  max:=0;

  maxi:=1;

  pred:=0;

  s:=0;

{просматриваем массив в поиска максимального элемента. Сначала просматриваем

массив от начала до конца, потом от следующего за максимальным до конца и т.д.,

пока максимальный элемент не станет равным последнему}

  while maxi<n do

        begin

        max:=0;

          for j:=i to n do

          if a[j]>max then

                begin

                 max:=a[j];

                 maxi:=j;

                end;

          s:=s+max*(maxi-pred);

          pred:=maxi;

          i:=maxi+1;

        end;

{выводим полученную сумму произведений в выходной файл}

 writeln(f2,s);

 close(f1);

 close(f2);

end.
Тесты к задаче
	№ теста
	_1.in
	_2.in
	_3.in
	_4.in
	_5.in
	_6.in

	Input.in
	5

73

31

96

24

46
	10

1

2

3

4

5

6

7

8

9

10
	10

10

9

8

7

6

5

4

3

2

1
	17

123

180

100

5

33

16

119

125

121

164

55

146

153

15

61

178

136
	25

90

190

46

1

124

99

149

26

80

132

53

4

151

45

52

196

126

148

67

153

47

17

155

1

106
	50

64

83

392

305

267

77

236

143

374

47

262

151

43

85

244

93

21

259

110

213

52

242

36

388

31

326

327

216

207

142

155

179

368

253

322

122

24

331

113

197

99

102

9

314

360

168

134

244

95

270

	Output.out
	380
	100
	55
	2988
	4433
	18306


2. Газон(100 баллов)
У Ивана Ивановича (И.И.) есть любимый газон – часть плоскости, в каждой точке с целочисленными координатами которой растет некоторое лекарственное растение. В выходной И.И. собрал лекарственные травы с прямоугольного участка (включая его границы), стороны которого параллельны осям координат, а координаты противоположных вершин (х1, у1), (х2,у2). Затем И.И. включил поливальную установку, которая размещена в точке с целочисленными координатами (х3,у3) и имеет радиус действия струи R (целое число).


Требуется написать программу для определения количества лекарственных растений, которые были одновременно собраны и политы.

Формат ввода:

В текстовом файле GASON.IN содержится: 

в первой строке через пробел целые числа
 х1,у1,х2,у2 (-10000
[image: image5.wmf]£

 х1,у1,х2,у2
[image: image6.wmf]£

 10000); 
во второй строке через пробел целые числа 
х3,у3, R (-10000
[image: image7.wmf]£

 х3, у3 
[image: image8.wmf]£

 10000, 1
[image: image9.wmf]£

 R
[image: image10.wmf]£

10000).

Формат вывода:

В единственной строке текстового файла GASON.OUT должно

 содержаться количество лекарственных растений, которые 

были и собраны, и политы.

	Пример входного файла
	Пример выходного файла
	Комменарий

	1  4  6  8

5  4  2 
	14
	См.Рисунок


Идея решения

Нужно только вспомнить, что в круг радиуса R и центром в точках (Х3,Y3) попадут все точки (Х,Y), для которых будет выполняться неравенство (X-X3)2+(Y-Y3)2
[image: image11.wmf]£

R2 . Если провести простой перебор, то уже 7 тестов из 10 легко пройдут. Не укладываемся во время в том случае, когда круг оказывается внутри  прямоугольника и имеет  радиус значительно меньших размеров, чем сам прямоугольник. Тогда имеет смысл сузить размеры прямоугольника до границ круга и проблема окажется решенной.
Текст программы
Program Gazon;

var
f1,f2:text;

x1,y1,x2,y2,x3,y3,r,minx,maxx,miny,maxy,I,j,sum:longint;

Begin
assign(f1,'_10.in’);reset(f1);

assign(f2,’gason.out’);rewrite(f2);

{читаем исходные данные из файла}

read(f1,x1);read(f1,y1);

read(f1,x2);read(f1,y2);

readln(f1);

read(f1,x3);read(f1,y3);

read(f1,r);

{определяем координаты левой нижней вершины

 прямоугольника и правой верхней}

if x1<x2 then begin  minx:=x1; maxx:=x2; end
         else begin  minx:=x2; maxx:=x1; end;

if y1<y2 then begin  miny:=y1; maxy:=y2; end

         else begin  miny:=y2; maxy:=y1; end;

{сужаем размеры прямоугольника до границ круга}

x31:=x3-r;

x32:=x3+r;

y31:=y3-r;

y32:=y3+r;

if minx<x31 then minx:=x31;

if maxx>x32 then maxx:=x32;

if miny<y31 then miny:=y31;

if maxy>y32 then maxy:=y32; 
sum:=0;

r:=sqr(r);

{просматриваем все точки нового прямоугольника и

проверяем их на принадлежность к кругу}

for i:=minx to maxx do

for j:=miny to maxy do

 if sqr(i-x3)+sqr(j-y3)-r<=0 then

                              sum:=sum+1;

{выводим полученную сумму в результирующий файл}

writeln(f2,sum);

close(f1);
close(f2);

End.
Тесты к задаче
	№ теста
	Input.in
	Output.out

	_1.in
	0 0 1 1

3 4 5
	4

	_2.in
	1 1 0 0

-7 0 8
	3

	_3.in
	-10 12 0 0

-5 6 7
	131

	_4.in
	-980 515 -1000 500

-996 504 5
	79

	_5.in
	1280  4270 1200 4400

1234 4321 50
	6989

	_6.in
	-90 -35 35 5

0 0 40
	2762

	_7.in
	9000 9000 8000 8000

9000 8000 1300
	988382

	_8.in
	9900 9900 10000 10000

9811 9801 133
	0

	_9.in
	-9814 -8235 7189 6204

5463 -1415 566
	1006369

	_10.in
	-10 -10000 9878 9836

10000 -100 10000
	154528300


3. Реклама (100 баллов)
Световое табло на стадионе имеет вид прямоугольника, разбитого на n строк и m столбцов. В каждой ячейке табло может отображаться определенный цвет. Возможные цвета кодируются так:

0 – если ячейка могла отображать только черный цвет;

1 – если ячейка может отображать только черный и синий цвета;

2 – если ячейка может отображать только черный и зеленый цвета;

3 – если ячейка может отображать только черный, зеленый и синий цвета;

4 – если ячейка может отображать только черный и красный цвета;

5 – если ячейка может отображать только черный, красный и синий цвета;

6 – если ячейка может отображать только черный, красный и зеленый цвета;

7 – если ячейка может отображать черный, красный, зеленый и синий цвета.


Требуется определить, можно ли отобразить на табло заставку рекламы без искажений, если задано ее описание с использованием четырех цветов: красного ®, зеленого (G), синего (B), черного (W).


Будем считать, что заставка отображается без искажений, если ее описание в каждой ячейке (I,j) соответствует описанию табло в этой ячейке.

Формат ввода:

В текстовом файле RECLAMA.IN записаны: 

в первой строке – целые числа n и m через пробел (1
[image: image12.wmf]£

 n, m
[image: image13.wmf]£

 100). В каждой из следующих n строк записаны по m символов – описание заставки. В следующих  n строках записано описание табло (n строк по m чисел через пробел).

Формат вывода:

Программа должна записать в текстовый файл  RECLAMA.OUT количество ячеек табло, в которых реклама будет отображаться с искажением. 

	Пример входного файла
	Пример выходного файла
	Комменарий

	  3 3

WGB 

RWB

RGW

0  1  2

3  4  5

6  7  0
	3
	Искажения в трех ячейках:
(1,2), (1,3), (2,1)

	2  3
RGB

WGW

6  7  5

7  7  7


	0
	Искажений нет


Идея решения

Считываем из исходного файла данные в два массива. Первый массив A– одномерный строковый и содержит n элементов по m символов. Второй массив B – двумерный целочисленный, состоящий из  n строк и  m столбцов, элементами которого являются целые числа от 0 до 7. Просматриваем массив В по строкам и сравниваем код цвета в J-ом столбце с описанием J-ого символа  I-ого элемента массива А. В случае, если проверяемым кодом b[i,j] нельзя описать  J-ый символ  I-ого элемента массива А a[i][j], делаем вывод, что наблюдается  искажение и какой-то счетчик SUM увеличиваем на 1. Таким образом, после окончания просмотра массива В в переменной SUM будет получено количество ячеек табло, в которых реклама будет отображаться с искажением. 

Текст программы 
Program Reclama;

 var

   f1,f2:text;

   a:array[1..150] of string;

   b:array[1..150,1..150] of integer;

   n,m,i,j,sum:longint;

   t:string;

Begin

 assign(f1,'reclama.in');reset(f1);

 assign(f2,'reclama.out');rewrite(f2);

 read(f1,n);read(f1,m);readln(f1);

 for i:=1 to n do readln(f1,a[i]);

 for i:=1 to n do

   begin

     for j:=1 to m do

       read(f1,b[i,j]);

     readln(f1);

   end;

 sum:=0;

 for i:=1 to n do

   for j:=1 to m do

{*}   begin

        t:=a[i];

        if b[i,j]=0 then if t[j]<>'W' then sum:=sum+1;


if b[i,j]=1 then if (t[j]<>'W')and(t[j]<>'B') then sum:=sum+1;


if b[i,j]=2 then if (t[j]<>'W')and(t[j]<>'G') then sum:=sum+1;


if b[i,j]=3 then if (t[j]<>'W')and(t[j]<>'G')and (t[j]<>'B') then sum:=sum+1;


if b[i,j]=4 then if (t[j]<>'W')and(t[j]<>'R') then sum:=sum+1;


if b[i,j]=5 then if (t[j]<>'W')and(t[j]<>'R')and(t[j]<>'B') then sum:=sum+1;


if b[i,j]=6 then if (t[j]<>'W')and(t[j]<>'R')and(t[j]<>'G') then sum:=sum+1;

{*}    end;

 writeln(f2,sum);

 close(f1);close(f2);

End.
Тесты к задаче
	№ теста
	_1.in
	_2.in
	_3.in
	_4.in
	_5.in

	Input.in
	4 1

R

R

B

G

5

4

1

3
	6 4

WBGB

GGRR

WGRW

WRWB

RWWB

GWRR

3 2 3 6

2 3 0 7

4 5 0 0

1 3 5 3

1 5 3 2

3 3 5 3
	9 6

RBBGRW

BBWRWB

BWGRGW

WRRWRW

RGBRGG

BGWWBR

GGGWBW

GRWWBG

WRGRRB

4 6 4 2 2 2

2 2 2 3 2 3

4 4 0 5 6 3

1 2 1 2 6 3

3 2 0 3 3 3

5 1 5 0 6 1

5 2 3 0 6 0

1 4 1 3 3 3

2 4 1 4 2 2
	18 8

WRRWWRGB

RBRWBRGB

WWRWBWBB

GGGGBBWW

GBGWGGRG

WGGRGGGR

BBRGWGGG

WBRRWBWR

RWBRWRGR

GGGGGBRB

WRGRGBGR

WGWBBWGG

GGWWWWBR

RGRWGRWG

RGRWWWRR

RGRGWGBW

GGGGBRGB

RRGGGRWG

3 2 4 3 1 1 2 2

3 3 2 6 0 3 4 5

5 7 5 3 0 6 3 5

3 4 1 2 4 7 0 6

6 4 5 5 6 7 5 2

4 5 5 7 2 4 1 6

1 2 7 3 5 4 4 3

6 6 0 7 1 2 0 3

5 4 6 2 1 5 4 5

7 4 3 2 0 1 7 5

4 1 0 1 6 4 3 2

1 4 4 4 1 5 3 5

3 1 1 3 6 6 3 6

4 2 3 1 3 5 1 1

1 4 7 2 6 6 3 7

5 0 6 6 1 1 1 1

5 1 7 7 6 6 6 7

0 0 5 2 3 3 7 1
	13 12

GBGWGWBBRGRW

WBRRRWRBRGGG

RWGGWGRBWBRG

BBGGRBWWWGRG

BWBWWWGRRGBW

WWRBRWWRRWBW

RWWRGWBWGGWG

BRRWBRGBWWRW

WGBGGRRRBGWW

WRGWWGRWGRBR

RRRWGGWBGBBG

RWRGWBGRWGRW

BBRBRWBWBBGG

0 0 7 6 0 4 0 3 5 1 4 1

1 3 4 7 5 2 2 3 0 3 2 5

2 7 6 3 0 2 6 7 5 6 2 4

6 7 5 1 4 3 3 0 7 5 1 2

4 0 7 1 5 5 7 0 1 1 6 1

1 3 1 3 7 0 1 0 1 4 0 3

6 5 2 4 0 1 4 6 2 7 7 1

3 1 2 0 4 3 7 4 2 0 2 3

3 3 1 0 2 3 6 0 2 4 0 6

5 2 0 7 1 0 0 6 0 1 0 4

1 6 5 7 7 1 3 1 0 2 1 4

0 1 0 4 7 3 1 3 7 4 3 5

6 5 1 7 3 3 0 1 2 0 1 5

	Output.out
	0
	9
	22
	54
	67


4. Интересные числа (100 баллов)
Требуется написать программу, которая подсчитает количество нечетных квадратных чисел на заданном числовом промежутке [a,b] (целое число называется квадратным, если квадратный корень из него является целым числом, например 9, 16, 100 и т.д.)

Формат ввода:

С клавиатуры вводятся через пробел целые числа a и b (1
[image: image14.wmf]£

а, b
[image: image15.wmf]£

2000000000).
Формат вывода:

Программа должна вывести на экран количество нечетных квадратных чисел, которые содержатся на заданном отрезке.

	Пример ввода
	Пример вывода
	Комменарий

	15  20
	0
	Нет ни одного нечетного числа

	10  178  
	5
	Это числа: 25, 49, 81, 121, 169


Идея решения

Прочитав условия олимпиадных задач, большинство участников олимпиады сразу принялись за решение этой задачи. Казалось, решение просто лежит на поверхности. 
Организуем циклом FOR просмотр заданного промежутка и в цикле:

1. проверям число на нечетность

2. в случае, если оно нечетное, проверяем, является ли оно квадратным, т.е. квадратный корень из него является целым числом, и если это так то увеличиваем счетчик SUM на 1. (Program Chisla_3_1)
Но если приступить к решению напрямую, особо не размышляя, то на последних 3-ех тестах компьютер не сможет уложиться в предоставленное ему время. Такая ситуация возникает из-за слишком большого числа переборов, если согласно входным данным нужно перебирать числа от 1 до 2000000000. 

Если немножко поразмышлять, можно начать просмотр промежутка с нечетного числа и шагом 2, но это не приведет к улучшению алгоритма, если не изменить тело цикла.( Program Chisla_3).

Поскольку нам нужны нечетные полные квадраты, то немного еще поразмышляв, приходим к выводу, что они могут получиться только из нечетных чисел. Вычисляя количество полных нечетных квадратов нужно смотреть не исходный промежуток, а получить новый. Для этого нужно извлечь корни из старых границ диапазона и округлить их в нужную сторону. Левой границу обязательно сдвинуть вправо до нечетного числа. И, не думая больше, циклом с шагом 2 перебрать весь новый полученный промежуток и вычислить нужное количество полных нечетных квадратов. (Program Chisla_2)
И, наконец, самое продвинутое решение, в котором цикл использовать вовсе не нужно. Поступаем как в предыдущем решении, только правую границу тоже двигаем влево до первого нечетного числа. Нужное количество полных нечетных квадратов вычисляем просто по формуле (конец-начало) div 2+1(Program Chisla_1)
Текст программы
Program Chisla_3_1;

   var
      a,b,sum,i:longint;

      j:real;

Begin

{вводим числовой промежуток}

 readln(a,b);

 sum:=0;

{просматриваем весь промежуток от a  до b}

for i:=a to b  do

   {проверяем, является ли число нечетным}

     if i mod 2 =1 then

     begin

      {вычисляем корень из числа}

      j:=sqrt(i);

   {проверям является ли число полным квадратом}

      if j=round(j) then sum:=sum+1;

   {берем следующее число из промежутка}

   end;

{выводим результат на экран}

 writeln(sum);

End.

*************************************************************************
Program Chisla_3;

   var

      a,b,sum:longint;

      j:real;

Begin

{вводим числовой промежуток}

 readln(a,b);

{если левая граница оказалась четной, увеличиваем ее на 1}

 if a mod 2=0 then a:=a+1;

 sum:=0;

{просматриваем весь промежуток с шагом 2,

т.к. нужны только нечетные квадраты}

 while a<=b do

   begin

   {вычисляем корень из числа}

      j:=sqrt(a);

   {проверям является ли число полным квадратом}

      if j=round(j) then sum:=sum+1;

   {берем следующее число из промежутка}

      a:=a+2;

   end;

{выводим результат на экран}

 writeln(sum);

End.
*************************************************************************

Program Chisla_2;

 var

    a,b,max,min,sum,i,j:longint;

Begin

{вводим числовой промежуток}

 readln(a,b);

{определяем левую и правую границы промежутка}

 if a>b then begin

            max:=a;

            min:=b;

            end

        else begin

            max:=b;

            min:=a;

            end;

{вычисляем корень из левой границы и округляем}

 i:=round(sqrt(min));

{если квадрат новой левой границы i оказался меньше

старой левой границы min, то новую левую границу увеличиваем на 1 }

 if sqr(i)<min then inc(i);

{если новая левая граница оказалась числом четным,

 то увеличиваем ее на 1}

 if i mod 2=0 then inc(i);

{вычисляем корень из правой границы и округляем}

 j:=round(sqrt(max));

{если квадрат новой правой границы j оказался больше

старой правой границы max, то норвую правую границу уменьшаем на 1 }

 if sqr(j)>max then dec(j);

{пройдя цикл от i до j шагом 2 подсчитываем количество

нечетных чисел}

 while i<=j do

  begin

   sum:=sum+1;

   i:=i+2;

  end;

{выводим результат на экран}

 writeln(sum);

End.

*************************************************************************

Program Chisla_1;

 var

    a,b,max,min,sum,i,j:longint;

Begin

{вводим числовой промежуток}

 
readln(a,b);

{определяем левую и правую границы промежутка}

 
if a>b then 
 begin

           
  max:=a;

            
  min:=b;

            
 end

        
else 
 begin

                 max:=b;

                 min:=a;

                end;

{вычисляем корень из левой границы и округляем}

 
i:=round(sqrt(min));

{если квадрат новой левой границы i оказался меньше

старой левой границы min, то норвую левую границу увеличиваем на 1 }

 
if sqr(i)<min then inc(i);

{если новая левая граница оказалась числом четным,

 то увеличиваем ее на 1}

 
if i mod 2=0 then inc(i);

{вычисляем корень из правой границы и округляем}

 j:=round(sqrt(max));

{если квадрат новой правой границы j оказался больше

старой правой границы max, то норвую правую границу уменьшаем на 1 }

 
if sqr(j)>max then dec(j);

 
if j mod 2 =0 then dec(j);

{просто вычисляем количество нужных нам чисел по формуле:

(конец-начало)/2+1)}

  
 sum:=(j-i) div 2+1;

{выводим результат на экран}

 
writeln(sum);

End.

Тесты к задаче
	№ теста
	Ввод
	Вывод
	Баллы

	1
	1 18
	2
	10

	2
	8 1576
	19
	10

	3
	1456 2041
	4
	10

	4
	4866 8391
	11
	10

	5
	8785 8999
	0
	10

	6
	45 35609
	91
	10

	7
	7 100334
	157
	10

	8
	962784710 1954550455
	6591
	10

	9
	13446 1847581758
	21434
	10

	10
	1 2000000000
	22361
	10


Задачи, подводящие к решению некоторых рассмотренных задач.
 Вспомогательные задачи к задаче «Газон»

1. Дан круг с центром в точке (x0,y0) и радиусом r, и даны координаты n точек. Напишите программу для подсчета количества точек, которые попали в круг или на его границу. 

Входные данные:

x0 y0 r 
далее идут координаты n точек 

Выходные данные:

kol - количество точек, попавших в круг 

Пример ввода:

0 0 1 
4 
0.5 0.5 
3 1 
1 0 
-5 -8 

Пример вывода:

2

2. Дано кольцо с центром в точке (x0,y0) и ограниченное окружностями радиусов r1 и r2 (r1 меньше r2), и даны координаты n точек. Напишите программу для подсчета количества точек, которые попали в кольцо (границу не учитывать). 
Входные данные:

x0 y0 r1 r2 
далее идут координаты n точек 

Выходные данные:

kol - количество точек, попавших в кольцо 

Пример ввода:

0 0 1 4 
5 
0 0 
1 0 
0 4 
2 2 
10 15 

Пример вывода:

1

Вспомогательные задачи к задаче «Баобаб»

1. Есть два массива из n чисел: a и b. Напишите программу, которая находит сумму произведений i-хы элементов массивов a и b: 

Sum=a[1]*b[1]+a[2]*b[2]+...+a[i]*b[i]+...+a[n]*b[n].

Формат ввода:

N - количество чисел в массивах, N<=10; 

a[1] b[1] 

a[2] b[2] 

... 

a[n] b[n], где a[i] - i-ое число в массиве a, b[i] - i-ое число в массиве b.

Формат вывода:

Sum - сумма произведений a[i]*b[i].

Пример ввода:

5 

1 1 

2 2 

3 3 

4 4 

5 5

2. У Васи имеется N денежных купюр разного достоинства. Он решил купить себе футбольный мяч, но не знает, хватает ли у него денег. Помогите Васе, написав программу, которая по заданному количеству купюр и их достоинству определяет, сколько всего денег у Васи. 

Формат ввода:

N — количество денежных купюр у Васи.

A1 — достоинство 1-ой купюры.

A2 — достоинство 2-ой купюры.

...

An — достоинство n-ой купюры. 

Формат вывода:

K — количество денег у Васи. 

Пример ввода:

5

10

20

10

5

3

Пример вывода:

48
3. Семья из N сусликов запасла на зиму M кг зерна. Известно, сколько граммов зерна ест каждый суслик в день. Определить, сколько граммов зерна останется в конце зимы, если зимовка длится T дней.

Ограничения:

0<=N<=100;

0<=A[i]<=100;

0<=M<=2000;

0<=T<=200; 

Формат ввода:

N M T

A[1]

...

A[N]

Где A[i] — количество зерна в граммах, которое съедает i-ый суслик в день.

Формат вывода:

G — оставшееся количество зерна в граммах. Если зерна на зиму не хватает, вывести -1.

Пример ввода:

2 1 150

3

2  

Пример вывода:

250 

Пример ввода:

5 14 5

200

400

600

800

1000  

Пример вывода:

-1 
4. Скоро будут проходить петушиные бои. В них хотят принять участие K петухов из различных стран мира. Судьи должны знать, сколько петухов в одной из категорий. Причем в этой категории вес самого легкого не меньше N кг, а максимальный не более M кг. Помогите им вычислить, сколько петухов в данной категории.

Ограничения

0<=N<=M<=1000

0<=A[I]<=1000

0<=K<=5000
Входные данные:

K — Количество петухов.

A[1] — Вес первого петуха

A[2] — Вес второго петуха

...

A[K] — Вес K-ого петуха.

N M — Вес самого легкого и тяжелого петуха.
Выходные данные:

Kol — Количество петухов в данной категории

Пример ввода:

4

1

2

3 

4 

2 3

Пример вывода:

2

5. Белоснежка шила одежду для семи гномов. У неё осталось немного ткани, и она не знала, хватит ли её, для того чтобы сшить что-нибудь для самого маленького гнома. Напишите программу, которая находит рост самого маленького из семи гномов.

Входные данные:

a[1] a[2] a[3] a[4] a[5] a[6] a[7] — массив ростов (10<=a[i]<=250) 

Выходные данные:

min — рост самого маленького гнома.

Пример ввода:

80

90

100

15

50

69

72

Пример вывода:

15

